Osnovne škole

Zadaci za državno natjecanje iz programskog jezika

LOGO 2000. godine
1. zadatak
20 bodova

Napišite proceduru LEPEZA :n :d :k :r :b koja će nacrtati niz mnogokuta kao na slici. Svi mnogokuti imaju :n stranica duljine :d. Procedura mora nacrtati :b mnogokuta. Svi mnogokuti se počinju crtati iz iste točke (vidi sliku: donji lijevi vrh mnogokuta). 

Drugi mnogokut je zaokrenut na desno u odnosu na prvi mnogokut za kut rotacije :k. Svaki sljedeći mnogokut je zarotiran u odnosu na prethodni mnogokut za prethodni kut rotacije podijeljen s :r.

Varijabla :n je veća ili jednaka jedan. Varijable :d, :k i :b su veće od nule ili jednake nuli. Varijabla :r je veća od nule.
Primjer (vidi sliku): CS LEPEZA 4 200 20 2 4
Program snimite na disketu pod imenom LEPEZA.LGO 
2. zadatak
40 bodova

Napišite proceduru PIRAMIDA :n :d :k koja će nacrtati piramidu kao na slici. Varijabla :n je broj kvadrata osnovice piramide odnosno njenog najdonjeg reda, a varijabla :d je duljina stranice kvadrata. Svi kvadrati su isti. Svaki sljedeći red prema vrhu ima jedan kvadrat manje od reda ispod njega. Red na vrhu ima jedan kvadrat.

Početak prvog reda je u središtu ekrana odnosno donji lijevi vrh prvog kvadrata prvog reda je u središtu ekrana. Prvi red je pod kutem :k / 2 od zamišljene horizontalne linije (vidi sliku).

Svaki sljedeći red započinje na polovici zadnjeg kvadrata prethodnog reda. Redovi su pod kutem :k kao što se vidi na slici. 

Varijable su veće od nule ili jednake nuli.

Primjer (vidi sliku): PIRAMIDA 4 30 45

Program snimite na disketu pod imenom PIRAMIDA.LGO

3. zadatak
60 bodova
Napišite proceduru UREDI :l koja će urediti listu :l na taj način da na početak liste stavi riječi koje se nalaze u listi, a na kraj liste stavi liste koje se nalaze u listi. Redoslijed riječi mora biti onaj kojim su te riječi upisane u listu i isto tako redoslijed lista mora biti onaj kojim su te liste upisane u listu. 

Na opisan način mora biti uređena i svaka lista koja se nalazi unutar neke liste. 

Uređenu listu procedura vraća kao svoj rezultat.

Primjer: 
    UREDI [sandra [[1 3] goran] marek [renata [2 [] 4] 5]]

Result: [sandra marek [goran [1 3]] [renata 5 [2 4 []]]]

Program snimite na disketu pod imenom UREDI.LGO

4. zadatak
80 bodova

Napišite proceduru DANI :dat1 :dat2 koja će izračunati koliko ima dana između datuma :dat1 i datuma :dat2 i te dane vratiti kao svoj rezultat. Za prijestupne godine uzmite godine koje su djeljive s četiri. 

Datumi :dat1 i :dat2 su riječi sljedećeg formata: broj do prve točke je dan, broj između prve i druge točke je mjesec i broj nakon druge točke je godina. Svi ti brojevi su mogući datumi, pa ne morate provjeravati njihovu ispravnost. Dani i mjeseci mogu, ali ne moraju imati vodeće nule.

Primjeri datuma: “01.02.1999 ,  “1.3.900 , “21.4.5000 , “31.12.1972
Primjer: 
DANI “16.2.1972 “6.4.1974

Result: 780

Program snimite na disketu pod imenom DANI.LGO

Osnovne škole

Test primjeri za državno natjecanje iz programskog jezika LOGO

2000. godine

Testiranje 1. zadatka


POZIV PROCEDURE
REZULTAT
BODOVI 1
BODOVI 2

1.
CS LEPEZA 4 100 45 1 0
Ništa se ne iscrtava
+4
+3

2.
CS LEPEZA 4 100 45 1 1
Slika 1A
+4
+3

3.
CS LEPEZA 4 100 45 1 8
Slika 1B 
+4
+3

4.
CS LEPEZA 4 100 0 1 8
Slika 1A
+4
+3

5.
CS LEPEZA 6 100 180 3 3
Slika 1C
+4
+3

NAJVEĆI MOGUĆI BROJ BODOVA
20

BODOVI 1: 
Rezultati su isti kao u test primjerima.

BODOVI 2: 
Rezultati su isti kao u test primjerima, ali je došlo do greške u programu ili je program ušao u beskonačnu petlju.

Napomena: Za jedan test primjer mogu se dobiti bodovi iz samo jednog seta bodova (ili BODOVI 1 ili BODOVI 2). 

Slika 1A
Slika 1B


Slika 1C

Testiranje 2. zadatka


POZIV PROCEDURE
REZULTAT
BODOVI 1
BODOVI 2
BODOVI 3

1.
CS PIRAMIDA 0 100 0
Ništa se ne iscrtava
+5
+4
-

2.
CS PIRAMIDA 1 100 0
Slika 2A
+7
+5
-

3.
CS PIRAMIDA 1 100 90
Slika 2B
+7
+5
-

4.
CS PIRAMIDA 2 100 0
Slika 2C
+7
+5
+3

5.
CS PIRAMIDA 2 80 180
Slika 2D
+7
+5
+3

6.
CS PIRAMIDA 7 10 90
Slika 2E
+7
+5
+3

NAJVEĆI MOGUĆI BROJ BODOVA
40

BODOVI 1: 
Rezultati su isti kao u test primjerima.

BODOVI 2: 
Rezultati su isti kao u test primjerima, ali je došlo do greške u programu ili je program ušao u beskonačnu petlju.

BODOVI 3: 
Ne iscrtavaju se svi redovi, ali mora biti iscrtan pravilno (:n kvadrata stranice duljine :d) bar prvi red.

Napomena: Za jedan test primjer mogu se dobiti bodovi iz samo jednog seta bodova (ili BODOVI 1 ili BODOVI 2 ili BODOVI 3). 

  Slika 2A
 Slika 2B


 Slika 2C
  Slika 2D


  Slika 2E

Testiranje 3. zadatka


POZIV PROCEDURE
REZULTAT
BODOVI 1
BODOVI 2

1.
UREDI [1 2 3]
[1 2 3]
+12
+8

2.
UREDI [[a b] 1 2 3]
[1 2 3 [a b] ]
+12
+8

3.
UREDI [[a b] 1 [m] 2 [n] 3]
[1 2 3 [a b] [m] [n] ]
+12
+8

4.
UREDI [[a [x] b] 1 [[x] m] 2 [[x] n] 3]
[1 2 3 [a b [x] ] [m [x] ] [n [x] ] ]
+12
+8

5.
UREDI [ [ [ [ [ ] 4] 3] 2] 1]
[1 [2 [3 [4 [ ] ] ] ] ]
+12
+8

NAJVEĆI MOGUĆI BROJ BODOVA
60

BODOVI 1: 
Rezultati su isti kao u test primjerima.

BODOVI 2: 
Rezultati su isti kao u test primjerima, ali je došlo do greške u programu ili je program ušao u beskonačnu petlju.

Napomena: Za jedan test primjer mogu se dobiti bodovi iz samo jednog seta bodova (ili BODOVI 1 ili BODOVI 2)
Testiranje 4. zadatka


POZIV PROCEDURE
REZULTAT
BODOVI 1
BODOVI 2

1.
DANI
 “01.01.2000
 “10.01.2000

9
+10
+7

2.
DANI
 “1.1.2000
 “1.1.2000

0
+10
+7

3.
DANI
 “1.1.2000
 “1.2.2000

31
+10
+7

4.
DANI
 “1.2.2000 
 “1.3.2000

29
+10
+7

5.
DANI
 “1.1.1999 
 “1.1.2000

365
+10
+7

6.
DANI
 “16.2.2000
 “6.4.2000

50
+10
+7

7.
DANI
 “30.11.1995
 “8.1.2000

1500
+10
+7

8.
DANI
 “30.11.1995
 “8.1.1990

2152
+10
+7

NAJVEĆI MOGUĆI BROJ BODOVA
80

BODOVI 1: 
Rezultati su isti kao u test primjerima.

BODOVI 2: 
Rezultati su isti kao u test primjerima, ali je došlo do greške u programu ili je program ušao u beskonačnu petlju.

Napomena: Za jedan test primjer mogu se dobiti bodovi iz samo jednog seta bodova (ili BODOVI 1 ili BODOVI 2)
Rješenje 1. zadatka za PC LOGO

to lepeza :n :d :k :r :b


if :b = 0 then stop


repeat :n [fd:d rt 360/:n]


rt :k


lepeza :n :d :k/:r :r :b-1

end

Rješenje 2. zadatka za PC LOGO

to piramida :n :d :k


cs rt 90 lt :k/2


piramid :n :d :k "TRUE

end

to piramid :n :d :k :smjer


if :n = 0 then stop


repeat :n [red :d :smjer]


novi.red :d :smjer :k


piramid :n-1 :d :k (not :smjer)

end

to red :d :smjer


repeat 5 [fd:d if :smjer then lt 90 else rt 90] 


if :smjer then rt 90 else lt 90

end

to novi.red :d :smjer :k


if :smjer then lt 90 else rt 90


fd :d


if :smjer then lt 90 else rt 90


fd:d/2


if :smjer then rt:k else lt :k

end

Rješenje 3. zadatka za PC LOGO

to uredi :l


local "rijeci


local "liste


local "i


make "rijeci []


make "liste []


for "i 1 count:l \



[if word? item:i:l then \




make"rijeci lput item:i:l :rijeci \



else \




if list? item:i:l then \





make"liste lput uredi item:i:l :liste \



]





op se :rijeci :liste

end

Rješenje 4. zadatka za PC LOGO

to dani :dat1 :dat2


make"d 0 make"m 0 make"g 0


extract:dat1


make"d1:d make"m1:m make"g1:g


make"d 0 make"m 0 make"g 0


extract:dat2


make"d2:d make"m2:m make"g2:g


if :g2<:g1 then zamjena else \



if :g2=:g1 then if :m2<:m1 then zamjena else \




if :m2=:m1 then if :d2<:d1 then zamjena


make"dani 0


while [(or :d1<:d2 :m1<:m2 :g1<:g2)] [make"dani :dani + 1 provjera make"d1 :d1 + 1]


op:dani

end

to zamjena


make"pd :d1 make"pm :m1 make"pg :g1


make"d1 :d2 make"m1 :m2 make"g1 :g2


make"d2 :pd make"m2 :pm make"g2 :pg

end

to provjera


if :d1 = 28 then if (and :m1=2 not prestupna:g1) then IncM


if :d1 = 29 then if (and :m1 = 2 prestupna:g1) then IncM 


if :d1 = 30 then if (or :m1=2 :m1=4 :m1=6 :m1=9 :m1=11) then IncM


if :d1 = 31 then if (or :m1=1 :m1=3 :m1=5 :m1=7 :m1=8 :m1=10 :m1=12) then IncM

end

to prestupna :god


if (remainder :god 4) = 0 then op "TRUE else op "FALSE

end

to IncM


make"m1 :m1 + 1


make"d1 0


if :m1 = 13 then make"g1 :g1 + 1 make"m1 1

end

to extract :date


make"ind 1


for "i 1 count:date \



[if equal? item:i:date ". then make"ind :ind + 1 \



 else [if :ind = 1 then make "d lput item:i:date :d else \




  if :ind = 2 then make "m lput item:i:date :m else \




  if :ind = 3 then make "g lput item:i:date :g else \




 ] \



]

end

Rješenje 1. zadatka za UCB LOGO

to lepeza :n :d :k :r :b


if :b = 0 [stop]


repeat :n [fd :d rt 360 / :n]


rt :k


lepeza :n :d :k / :r :r :b - 1

end

Rješenje 2. zadatka za UCB LOGO

to piramida :n :d :k


cs rt 90 lt :k / 2


piramid :n :d :k "TRUE

end

to piramid :n :d :k :smjer


if :n = 0 [stop]


repeat :n [red :d :smjer]


novi.red :d :smjer :k


piramid :n - 1 :d :k (not :smjer)

end

to red :d :smjer


repeat 5 [fd :d ifelse :smjer [lt 90] [rt 90]] 


ifelse :smjer [rt 90] [lt 90]

end

to novi.red :d :smjer :k


ifelse :smjer [lt 90] [rt 90]


fd :d


ifelse :smjer [lt 90] [rt 90]


fd :d / 2


ifelse :smjer [rt :k] [lt :k]

end

Rješenje 3. zadatka za UCB LOGO

to uredi :l


local "rijeci


local "liste


local "i


make "rijeci []


make "liste []


for [i 1 count:l] ~



[ifelse word? item :i :l [make "rijeci lput item :i :l :rijeci] ~



 [if list? item :i :l [make "liste lput uredi item :i :l :liste]]]


op se :rijeci :liste

end

Rješenje 4. zadatka za UCB LOGO

to dani :dat1 :dat2


make "d 0 make "m 0 make "g 0


extract :dat1


make "d1 :d make "m1 :m make "g1 :g


make "d 0 make "m 0 make "g 0


extract :dat2


make "d2 :d make "m2 :m make "g2 :g


ifelse :g2 < :g1 [zamjena] ~



[if :g2 = :g1 [ifelse :m2 < :m1 [zamjena] ~




[if :m2 = :m1 [if :d2 < :d1 [zamjena]]]]]


make "dani 0


while [(or :d1 < :d2 :m1 < :m2 :g1 < :g2)] [make "dani :dani + 1 provjera make "d1 :d1 + 1]


op :dani

end

to zamjena


make "pd :d1 make "pm :m1 make "pg :g1


make "d1 :d2 make "m1 :m2 make "g1 :g2


make "d2 :pd make "m2 :pm make "g2 :pg

end

to provjera


if :d1 = 28 [if (and :m1 = 2 not prestupna :g1) [IncM]]


if :d1 = 29 [if (and :m1 = 2 prestupna :g1) [IncM]]


if :d1 = 30 [if (or :m1 = 2 :m1 = 4 :m1 = 6 :m1 = 9 :m1 = 11) [IncM]]


if :d1 = 31 [if (or :m1 = 1 :m1 = 3 :m1 = 5 :m1 = 7 :m1 = 8 :m1 = 10 :m1 = 12) [IncM]]

end

to prestupna :god


ifelse (remainder :god 4) = 0 [op "TRUE] [op "FALSE]

end

to IncM


make "m1 :m1 + 1


make "d1 0


if :m1 = 13 [make "g1 :g1 + 1 make "m1 1]

end

to extract :date


make "ind 1


for [i 1 count :date] ~



[ifelse equal? item :i :date ". ~



  [make "ind :ind + 1] ~



  [if :ind = 1 [make "d lput item :i :date :d] ~



   if :ind = 2 [make "m lput item :i :date :m] ~



   if :ind = 3 [make "g lput item :i :date :g] ~



  ] ~



]

end

